

```

; =====
;
;   Simulation of an axially loaded pile.
;
; =====

set logfile pal.log
set log on
;
new
title
Axially Loaded Pile Demonstration Problem
;
; =====
; Create a single soil block, set its material properties,
; specify boundary conditions, apply in-situ stresses,
; turn on gravity, and ensure that the gravity forces are
; in balance with the in-situ stresses.
;
gen zone brick size 7 7 7 edge=11
;
mod elastic
pro bulk=5e9 she=1e9 density=2000
;
fix z range z=(-0.01, 0.01)
fix x range x=(-0.01, 0.01)
fix x range x=(10.99, 11.01)
fix y range y=(-0.01, 0.01)
fix y range y=(10.99, 11.01)
;
init szz -2.20e5 grad=(0, 0, 2.0e4)
init sxx -1.32e5 grad=(0, 0, 1.2e4)
init syy -0.88e5 grad=(0, 0, 8.0e3)
;
set gravity=(0.0, 0.0, -10.0)
solve
;
; =====
; Create a pile in the center of the soil block; set the
; properties of the pile.
;
sel pile id=1 begin=(5.5, 5.5, 12.0) end=(5.5, 5.5, 4.0) nseg=8
;
sel pile prop Emod=8.0e10 Nu=0.30 XCArea=0.7854 &
              XCJ=9.82e-2 XCly=4.91e-2 XCly=4.91e-2 &

```

```

        Per=3.14 &
        CS_sK=1.3e11 CS_sCoh=0.0    CS_sFric=30.0 &
        CS_nK=1.3e09 CS_nCoh=1.0e10 CS_nFric=30.0 CS_nGap=off
;
; =====
; Specify and apply velocity to pile tip.
;
def initval
    push_vel = -1.0e-8
end
initval
;
cycle 1 ; force the local-nodal systems of all nodes associated with
        ; pile to be updated
sel node FIX z range cid=1
sel node INIT zVel=push_vel range cid=1
;
; =====
; Set up histories for monitoring model behavior
;
history nstep=10
hist id=1 unbal
hist id=2 sel node zdisp id=1
hist id=4 sel pileSEL Fx id=1
hist id=6 sel pileSEL Fx id=3
hist id=8 sel pileSEL Fx id=5
hist id=10 sel pileSEL Fx id=7
hist id=12 sel pileSEL Fx id=8
;
;
; =====
; Apply velocity to achieve total displacement of 40e-5 m
;
cycle 40000
;
; =====
; Save file and turn off logfile
;
sav pal.sav
set log off
;
; =====
return

```